

## ARTICLES FROM THE SCiP CONFERENCE

---

# Meta-analysis using linear mixed models

CHING-FAN SHEU

*DePaul University, Chicago, Illinois*

and

SAWAKO SUZUKI

*University of California, Berkeley, California*

Psychologists often use special computer programs to perform meta-analysis. Until recently, this had been necessary because standard statistical packages did not provide procedures for such analysis. This paper introduces linear mixed models as a framework for meta-analysis in psychological research, using a popular general purpose statistical package, SAS. The approach is illustrated with three examples, using SAS PROC MIXED.

Meta-analysis of psychological studies has often been restricted to presenting estimates of summary statistics (i.e., effect sizes, often using standardized mean differences, odds ratios, or correlation coefficients) or estimating (fixed) moderator effects on the basis of fixed-effects linear models. Fixed-effects analysis models the systematic between-study differences and assumes subject-level sampling error in the studies included in a meta-analytical research.

A random-effects framework, on the other hand, conceptualizes the current set of studies under consideration as a random sample picked from a larger population of studies. That is, each study-specific effect is sampled from the larger population of effects. Thus, each study has its own population effect, and an inference is made about the larger population of effects. There are two sources of variability in the random-effects framework: one that is due to the variability of the effect parameters, and another that is due to the sampling variability of experimental units (i.e., subjects) in studies. In other words, random-effects analysis takes into account the *true variance* (or the remaining unmeasured random effect between studies), in addition to the modeled between-study differences and the sampling error assumed in fixed-effects models.

Besides the theory-driven decisions meta-analysts make (Hedges & Vevea, 1998), practical reasons—the lack of specialized computer programs, such as HLM (Raudenbush, Bryk, & Congdon, 2000) or MLwiN (Rasbash et al., 2000), or the time and effort to become familiar with a new interface—discourage many from conducting random-effects analyses. (See Normand, 1995, for a comparative review of specialized meta-analytic software packages.)

We believe that there are additional advantages to using general purpose statistical software to perform random-effects analysis. Meta-analysis involves an array of data manipulation procedures, such as creating, combining, or summarizing data sets, as well as preparing reports of analytic results, and the advantage of having various data management capabilities in a single statistical package cannot be overstated. Furthermore, users of specialized multilevel programs may dismiss the fact that many meta-analytical procedures belong to a single class of statistical models, called the *mixed-effects linear model* (Ware, 1985).

Fitting linear mixed-effects models by using the statistical software S-plus (Pinheiro & Bates, 2000) is another option. However, we focus on using SAS for the purposes of this paper, since SAS syntax is relatively simple and the software is widely available and more familiar among psychologists. Lipsey and Wilson (2001) offer an SPSS macro to fit fixed- or random-effects models for meta-analysis, but not linear mixed-effects models.

SAS PROC MIXED, a built-in procedure of SAS that was designed to conduct mixed-effects analysis, provides researchers with an attractive alternative to conducting random-effects meta-analysis by using specialized software. The purpose of this paper is to present random-effects meta-analysis as a special case of mixed-effects linear models and to demonstrate the use of SAS PROC MIXED to fit such meta-analytic models. Because SAS is one of the commonly used standard statistical packages among psychological researchers, our approach offers not only conceptual generality, but also practical efficiency.

In their discussions of random-effects models for meta-analysis, Hedges and Olkin (1985) and Hedges and Vevea (1998) referred to SAS procedures for carrying out the computations. Unfortunately, however, these authors did not provide any SAS syntax that would have facilitated readers' conducting such procedures. A comprehensive guide to performing meta-analysis with SAS software is

---

This research was supported in part by award from the Paid-Leave Program, granted to C.-F.S. by the University Research Council at DePaul University. Correspondence concerning this article should be addressed to S. Suzuki, Graduate School of Education, University of California, 4511 Tolman Hall, Berkeley, CA 94720 (e-mail: sawako@uclink4.berkeley.edu).

provided by Wang and Bushman (1999). The present tutorial is intended to complement the former documentation by offering three examples of meta-analysis from the field of psychology, along with explanations and interpretations focusing primarily on the MIXED procedure.

In the next two sections, we formulate a random-effects model for normally distributed effect sizes and show that it is a special case of mixed-effects linear models. We also argue that random effects should be estimated in some meta-analysis research. The succeeding three sections contain examples illustrating our approach. The first example concerns studies of gender difference in field articulation ability (Hyde, 1981), which use mean differences as effect sizes. The second example illustrates how a mixed-effects model involving a study-level covariate reduces to a fixed-effects linear model, using a meta-analysis of studies on the effectiveness of two particular tests that measure cognitive impairment in elderly patients (Hasselblad & Hedges, 1995). Finally, a meta-analysis of studies on addiction intervention programs presents the analysis of a series of  $2 \times 2$  tables, using odds ratio (on a logarithmic scale) as the effect size measure (Haddock, Rindskopf, & Shadish, 1998).

### MIXED-EFFECTS LINEAR MODELS

Analysis of variance (ANOVA) and linear regression models are two examples of the general linear models familiar to psychologists. As will be illustrated below, the mixed-effects linear model represents an extension of such general linear regression models and the random-effects ANOVA models.

The general form of mixed-effects linear models is

$$Y = X\beta + Zb + e, \tag{1}$$

where  $Y$  is the  $(k \times 1)$  vector of summary statistics (effect sizes) from a number of  $k$ -related but independent studies,  $X$  ( $k \times p$ ) is the design matrix describing study characteristics (covariates) that influence fixed effects,  $\beta$  ( $p \times 1$ ) is the vector of fixed-effects parameters,  $Z$  ( $k \times q$ ) is another design matrix describing the covariates for the random effects,  $b$  ( $q \times 1$ ) is the vector of random effects or the residuals on the between-study level, and  $e$  ( $k \times k$ ) is the matrix of residuals on the within-study level.

The effect sizes are assumed to be normally distributed.<sup>1</sup> The random effect  $b$  has a multivariate normal distribution with a zero mean vector and a covariance matrix  $D$  ( $k \times k$ ). The chance error  $e$  is a multivariate normal with zero means and a covariance matrix  $R$  ( $k \times k$ ). The random effect  $b$  and the chance error  $e$  are assumed to be independent of each other. The within-study residuals are assumed to be normally distributed. Moreover, they are usually assumed to be distributed in an identical manner within each group. However, for a meta-analysis involving large sample sizes, the within-study variances can be considered known, and the covariance matrices  $R$  are specified as diagonal matrices with known sample variances of the study effect sizes on their diagonals.

In the simplest case, in which  $k$  number of primary studies in a meta-analysis have the same characteristics and the residuals on the between-study level are the only random effects, we obtain a random-effects one-way ANOVA model with fixed within-study variances. In terms of the mixed-effects formulation, this model can be written as

$$Y = \mu + b + e, \tag{2}$$

where  $Y$  is the  $k$ -vector of effect sizes observed from the studies and  $\mu$  ( $k \times 1$ ) is the overall average population effect.

The specification of study-specific chance error  $e$  is the same as above, and the random effect  $b$  has a normal distribution with zero means and a single variance parameter  $\tau$  for the  $(k \times k)$  variance-covariance matrix. Currently, a common method for estimating variance components (i.e.,  $\tau$ ) in general linear models is to use the so-called restricted maximum likelihood (REML). In simple terms, the REML estimates of  $\tau$  take into account the loss of degrees of freedom in the estimation of fixed effects. The reader is referred to Brown and Kempton (1994) for an accessible account of REML estimation and its applications. By default, SAS PROC MIXED estimates the covariance parameters with the REML method.

### WHY ESTIMATE RANDOM EFFECTS?

The key result of the previous random-effects model is that, given the observed effect sizes, the estimated overall average treatment effect ( $\mu$ ), and the between-study variance ( $\tau$ ), an estimate of the (true) study-specific effect size for the  $i$ th study is normally distributed, with a mean of

$$a_i \mu + (1 - a_i) y_i, \tag{3}$$

where  $a_i = s_i / v_i$  and  $v_i = (s_i + \tau)$ ,  $s_i$  is the known within-study variance for the  $i$ th study, and  $y_i$  is the observed effect size for the  $i$ th study. Thus, an estimate of the true study-specific effect size is a weighted average of the estimated overall effect size  $\mu$  (averaged over all the studies combined) and the observed effect size of a particular study. The weight is determined by the relative sizes of the between-study and the within-study variances. It is clear that the estimated overall average effect size  $\mu$  is a good summary measure for all the studies combined only if the between-study variance  $\tau$  is much smaller than the within-study variances.

It can be shown that when the between-study variance  $\tau$  is known, the maximum likelihood estimate of the overall average effect size  $\mu$  is

$$\frac{\sum_{i=1}^k \frac{y_i}{v_i}}{\sum_{i=1}^k \frac{1}{v_i}}$$

In other words,  $\mu$  is the weighted average of effect sizes with weights equal to the inverse of the sum of the variance

components. The fixed-effects approach widely used in psychological research assumes that the between-study variance  $\tau$  is zero and checks the assumption by the homogeneity test (Hedges & Olkin, 1985). In theory, however, the between-study variance is rarely zero and should be accounted for in the overall treatment effect size.

### EXAMPLE 1

#### Gender Differences in Field Articulation Ability

##### The Data

This example is taken from Hedges and Vevea (1998), who used Hyde's (1981) meta-analytic studies on cognitive gender differences to illustrate methods for meta-analysis. This particular meta-analysis included the results of 14 independent studies on gender differences in field articulation ability and used standardized difference of means as the effect size.

The sample sizes of the 14 studies ranged from 30 to 163, with a median sample size of 45. The variances of the differences of means (i.e., the effect sizes) can be estimated from the observed data, and the within-study variances are considered to be known for each study. This estimation method is reasonable because the sample sizes of the studies are quite large. A simple random-effects model (Equation 2) is used to analyze this data set.<sup>2</sup> The focus here is to assess the overall average gender effect from the 14 individual studies and to gauge the amount of variability among these studies. In other words, we wish to estimate the parameters  $\mu$  and  $\tau$  of  $b$  (the random effect) in Equation 2. The effect sizes of the 14 studies are the observed responses  $Y$ , and the variances of the effect sizes are known variances in the diagonal of the variance-covariance matrix  $R$ .

##### SAS Syntax

Listing 1 displays the SAS statements for the analysis of the gender difference data (Hyde, 1981). We assume that readers are familiar with the general implementation and execution of SAS programs, but those seeking for more details about the program are referred to the *SAS/STAT*

*User's Guide, Version 8* (SAS Institute, 1999). Details about fitting mixed-effects models are offered in *SAS System for Mixed Models* (Littell, Milliken, Stroup, & Wolfinger, 1996).

Under the DATA and INFILE statements, we name the data set *sexdif*, which is read in from the text file, *sexdif.asc*. As we see under the INPUT command, this data set consists of two columns, *study* and *diff*. The first column is the study identification number (from 1 to 14), and the second column is the observed difference in means (effect sizes) for each study.

The CLASS statement under PROC MIXED specifies categorical variable(s) (*study* in our case) not containing quantitative information. The MODEL, RANDOM, and REPEATED statements together specify the statistical model we are fitting onto the data set. The MODEL statement identifies our dependent variable and the fixed effects. In our model, the effect sizes (*diff*) are modeled by the fixed-effects of the intercept, which is implied by default. The RANDOM and REPEATED statements together estimate the random effects (*study*) and the between-study variance. We request SAS to include in the output the estimates, standard errors, *t* statistics, and *p* values for significance testing for each of the fixed (the average overall treatment) and random effects, using the SOLUTION options. The P option provides us with a table of predicted study-specific effect sizes and residuals.

We supply the initial estimates of the parameters under the PARMS statement.<sup>3</sup> However, there is no need to estimate the within-study variances, since they are known. Hence, the EQCONS option holds those parameters constant. On the other hand, we give the between-study variance parameter an initial value of .050, which is roughly the average of all the within-study variances. (By supplying a rough approximation as the starting value, we can facilitate the estimation process for SAS.) There is evidence that variance between the studies exists, or in other words, we find heterogeneity among the studies, and therefore the between-study variance cannot be ignored in this example. Such a parameter is estimated by using the default REML method.

##### SAS Output

Listing 2 displays output from the call to SAS PROC MIXED. The Solution for Fixed Effects provides the estimated overall average gender difference in field articulation ability and its standard error, based on the 14 studies. We see that the average effect size is approximately .5492, with a standard error of .0967. On the basis of these estimates, we can calculate the 95% confidence interval of the average effect size, which is between .3597 and .7387. Listed under the Covariance Parameter Estimates (REML) is the estimated residual on the between-study level, or the random effect.

### EXAMPLE 2

#### Cognitive Tests for Elderly

##### The Data

Hasselblad and Hedges (1995) introduced a meta-analysis of four studies on the effectiveness of two cognitive tests designed to detect mild cognitive impairment

##### Listing 1

##### SAS Codes for Example 1 (Hedges & Vevea, 1998; Hyde, 1981)

```
TITLE 'Gender Difference Studies';

DATA sexdif;
  INFILE 'sexdif.asc';
  INPUT study diff;

PROC MIXED DATA=sexdif;
  CLASS study;
  MODEL diff = / P SOLUTION;
  RANDOM study / SOLUTION;
  REPEATED / GROUP = study;

PARMS (.050)
      (.071)(.033)(.137)(.135)(.140)
      (.095)(.106)(.121)(.053)(.025)
      (.044)(.092)(.052)(.095) / EQCONS=2 to 15;

RUN;
```

**Listing 2**  
**SAS Output of Example 1**

Gender Difference Studies					
Covariance Parameter Estimates (REML)					
Cov Parm	Group	Estimate	Alpha	Lower	Upper
STUDY		0.05638090	0.05	0.0202	0.4668
DIAG	STUDY 1	0.07100000	.	.	.
DIAG	STUDY 2	0.03300000	.	.	.
DIAG	STUDY 3	0.13700000	.	.	.
.	.	.	.	.	.
DIAG	STUDY 14	0.09500000	.	.	.
Solution for Fixed Effects					
Effect	Estimate	Std Error	DF	t	Pr >  t
INTERCEPT	0.54921090	0.09672197	13	5.68	0.0001

in the elderly. The two cognitive tests were the Trailmaking Test and the Misplaced-Objects Task (MOT), and the goal of the analysis was to compare the average effectiveness of the two screening tests.

**SAS Syntax**

The SAS statements used to conduct this analysis are illustrated in Listing 3. The two types of cognitive tests (Trail and MOT) in this meta-analysis are included in the model as covariates (fixed effects) at the study level. Therefore, the variable (*cogtest*) is listed under the CLASS (because it is a categorical variable) and MODEL statements.

**SAS Output**

Listing 4 shows the SAS output to our previous set of statements. The Covariance Parameter Estimates indicate that the between-study variance estimate is equal to zero, which suggest that the random-effects model is identical to the fixed-effects model for this particular analysis. This does not necessarily imply that there is no variance between the four studies. Rather, the estimated value of the random effect was “set” to zero, because the residuals on the between-study level were very small or negligible, relative to the residuals on the within-study level.

As is displayed under the Solution for Fixed Effects, the overall estimated average effect size for the Trailmaking tests is 1.87, with a standard error of approximately 0.21. The average effect size for the MOT is 1.32 (1.87 – 0.55), and the standard error for the fixed effect of the MOT is about 0.27. An approximate 95% confidence interval for the difference in effectiveness of the two screening tests is given by

$$(1.87 - 1.32) \pm 1.96 * 0.27 = [0.02, 1.08].$$

We see that the mean effect size for the Trailmaking tests is significantly greater than that for the MOT.

**EXAMPLE 3**

**Psychosocial Treatments for Addiction**

**The Data**

Haddock et al. (1998) reported results from 24 studies on the effectiveness of psychosocial treatments for indi-

viduals who were addicted to alcohol (12 studies), to illegal drugs (5 studies), or to smoking cigarettes (7 studies). The raw data for each study are the number of successes over the total number of clients in each of the treatment and control groups. These authors used odds ratio (on the logarithmic scale) as the effect size measure.

Let  $r_c$  be the number of successes in the control group and  $r_t$  be the number of successes in the treatment group of a particular study; let  $n_c$  and  $n_t$  be the respective sample sizes of the two groups in the study. Then, the log odds ratio (i.e., the effect size) for this study is estimated by

$$\log (\{r_t(n_c - r_c)\} / \{r_c(n_t - r_t)\}).$$

When the sample size of a study is large, the distribution of log odds ratio approximates the normal distribution with a mean of

$$\log (\{p_t(1 - p_c)\} / \{p_c(1 - p_t)\});$$

where  $p_c$  is the probability of success in a control group and  $p_t$  is the probability of success in a treatment group of the same study. The (asymptotic) variance of the log odds ratio (see Agresti, 1996) is

$$\frac{1}{r_t} + \frac{1}{n_t - r_t} + \frac{1}{r_c} + \frac{1}{n_c - r_c}.$$

Haddock et al. (1998) specified a random-effects model on the log odds ratio for their meta-analysis:

$$LOR_i = \beta_0 + \beta_1 \text{Alcohol}_i + \beta_2 \text{Smoke}_i + b_i + e_i, \quad (4)$$

where  $LOR_i$  is the observed log odds ratio for the  $i$ th study,  $\text{Alcohol}_i$  and  $\text{Smoke}_i$  are the indicator variables for the  $i$ th study (e.g.,  $\text{Alcohol}_i$  is 1 if the  $i$ th study is an alcohol study, otherwise its value is 0),  $\beta_0$  is the average treatment effect size for illegal drug addiction,  $\beta_1$  is the average treatment effect size for alcohol above that of the drug studies,  $\beta_2$  is the average treatment effect size for smoking above that of the drug studies,  $b_i$  is the random

**Listing 3**  
**SAS Codes for Example 2 (Hasselblad & Hedges, 1995)**

```
TITLE 'Elderly Studies';

DATA elderly;
  INPUT study effsize cogtest $;
CARDS;
  1 1.75 Trail
  2 1.94 Trail
  3 1.34 MOT
  4 1.30 MOT
;

PROC MIXED DATA=elderly;
  CLASS study cogtest;
  MODEL effsize = cogtest / P SOLUTION;
  RANDOM study / SOLUTION;
  REPEATED / GROUP = study;

PARMS (.08)
      (0.1209)(0.07)(.048)(0.0757)
      / EQCONS=2 to 5;

RUN;
```

**Listing 4**  
**SAS Output of Example 2**

Elderly Studies						
Covariance Parameter Estimates (REML)						
Cov Parm	Group	Estimate				
STUDY		0.00000000				
DIAG	STUDY 1	0.12090000				
DIAG	STUDY 2	0.07000000				
DIAG	STUDY 3	0.04800000				
DIAG	STUDY 4	0.07570000				

  

Solution for Fixed Effects						
Effect	COGTEST	Estimate	Std Error	DF	t	Pr >  t
INTERCEPT		1.87033002	0.21055192	2	8.88	0.0124
COGTEST	MOT	-0.54585144	0.27148923	0	-2.01	.
COGTEST	Trail	0.00000000	.	.	.	.

effect or the residuals on the between-study level, and  $e_i$  is the residual on the within-study level. In terms of mixed-effect linear models, we write

$$LOR = X\beta + b + e, \tag{5}$$

where  $LOR$  is a  $24 \times 1$  vector of log odds ratios,  $X$  is a  $24 \times 3$  design matrix with 1s on the first column followed by two columns of values for the indicator variables,  $\beta$  is a  $3 \times 1$  vector of fixed-effects parameters,  $b$  is a multivariate normal with a mean vector 0 ( $24 \times 1$ ) and a variance-covariance matrix that equals  $\tau$  times a  $24 \times 24$  identity matrix, and  $e$  is also a multivariate normal with a mean vector 0 ( $24 \times 1$ ) and a variance-covariance matrix of a  $24 \times 24$  diagonal matrix with the known variances of the log odds ratios on the diagonals.

**SAS Syntax**

The SAS program file for fitting the model specified in Equation 5 is shown under Listing 5. These SAS codes basically follow the same syntax pattern as that in the first example, but two additions are noted. First, we added the COVTEST option under the MIXED procedure to request the printing of standard errors and test statistics for the variance and covariance parameters. However, these test results should be used with great care (see Verbeke & Molenberghs, 1997). Also, we now have the covariates at the study level. Since we have three types of studies—illegal drugs, alcohol, and smoking—studies concerning illegal drugs were arbitrarily chosen as the baseline (as expressed in Equation 4), and therefore the model included dummy codings of the other two variables, *alcoh* and *smoke*. These two variables were listed under the MODEL statement in the MIXED procedure to estimate the average treatment effect size of the drug studies ( $\beta_0$  in Equation 4) and the average treatment effect sizes of alcohol ( $\beta_1$ ) and smoking studies ( $\beta_2$ ) above that of the drug studies.

**SAS Output**

Output from PROC MIXED is shown in Listing 6. Under the Solution for Fixed Effects, we see that the average effect sizes for illegal drug, alcohol, and smoking cessa-

tion studies are approximately 1.2845, 0.0332 (1.2845 – 1.2513), and 0.3979 (1.2845 – 0.8867), respectively. It was found that psychosocial treatments appeared to be most effective for individuals addicted to illegal drugs and least effective for those individuals addicted to alcohol.

The estimated value of interstudy variance is 0.136, and its standard error estimate is 0.0968. If this estimate is significantly different from zero (unlike the present case), it indicates that the overall average effect size must be interpreted as the mean effect, rather than as the effect, disregarding the size of the within-study variances.

**SUMMARY**

Our three examples illustrated how to conduct a meta-analysis by using linear mixed-effects models in SAS. The first example contained 14 studies, across which we found heterogeneity. By fitting a random-effects model to the data, we were able to estimate the overall effect size, taking into consideration the between-study variance.

**Listing 5**  
**SAS Codes for Example 3**  
**(Haddock, Rindskopf, & Shadish, 1998)**

```
TITLE 'Addiction Studies';

DATA addict;
  INFILE 'addict.asc';
  INPUT study lor alcoh smoke;

PROC MIXED COVTEST DATA=addict;
  CLASS study;
  MODEL lor = alcoh smoke / P SOLUTION;
  RANDOM study / SOLUTION;
  REPEATED / GROUP = study;

PARMS (.50)
  (1.83)(.52)(1.03)(.86)(.55)
  (.19)(.39)(2.03)(.63)(.37)
  (.33)(.08)(.44)(.41)(.2)
  (.62)(.55)(.13)(.08)(.23)
  (.17)(.25)(.12)(.04) / EQCONS=2 to 25;

RUN;
```

**Listing 6**  
**SAS Output of Example 3**

Addiction Studies						
Covariance Parameter Estimates (REML)						
Cov Parm	Group	Estimate	Std Error	Z	Pr >  Z	
STUDY		0.13591875	0.9675427	1.40	0.1601	
DIAG	STUDY 1	1.83000000	.	.	.	
DIAG	STUDY 2	0.52000000	.	.	.	
DIAG	STUDY 3	1.03000000	.	.	.	
.	.	.	.	.	.	
DIAG	STUDY 24	0.04000000	.	.	.	

  

Solution for Fixed Effects						
Effect	Estimate	Std Error	DF	t	Pr >  t	
INTERCEPT	1.28453744	0.32745844	21	3.92	0.0008	
ALCOH	-1.25130325	0.39395937	0	-3.18	.	
SMOKE	-0.88667695	0.38059938	0	-2.33	.	

The second example illustrated how the mixed-effects linear model reduces itself to a fixed-effects model when the between-study variation is very small relative to the within-study variances. In some meta-analytic cases, it is not appropriate to report a single effect size to characterize all of the studies included in the analysis. Our last example showed how one could report different effect sizes for different study characteristics included in one meta-analysis by introducing covariates into the mixed-effects model. The three examples together speak to the advantages of modeling mixed-effects linear models to the data when conducting a meta-analysis.

#### REFERENCES

- AGRESTI, A. (1996). *An introduction to categorical data analysis*. New York: Wiley.
- BEGG, C. B. (1994). Publication bias. In H. Cooper & L. V. Hedges (Eds.), *The handbook of research synthesis* (pp. 399-409). New York: Russell Sage Foundation.
- BROWN, H. K., & KEMPTON, R. A. (1994). The application of REML in clinical trials. *Statistics in Medicine*, **13**, 1601-1617.
- DERSIMONIAN, R., & LAIRD, N. (1986). Meta-analysis in clinical trials. *Controlled Clinical Trials*, **7**, 177-188.
- HADDOCK, C. K., RINDSKOPF, D., & SHADISH, W. R. (1998). Meta-analysis using odds ratio as effect size measure: A primer on methods and issues. *Psychological Methods*, **3**, 339-353.
- HASSELBLAD, V., & HEDGES, L. V. (1995). Meta-analysis of screening and diagnostic tests. *Psychological Bulletin*, **117**, 167-178.
- HEDGES, L. V., & OLKIN, I. (1985). *Statistical methods for meta-analysis*. San Diego: Academic Press.
- HEDGES, L. V., & VEVEA, J. L. (1998). Fixed- and random-effects models in meta-analysis. *Psychological Methods*, **3**, 486-504.
- HYDE, J. S. (1981). How large are cognitive gender differences: A meta-analysis using omega and *d*. *American Psychologist*, **36**, 892-901.
- LIPSEY, M. W., & WILSON, D. B. (2001). *Practical meta-analysis*. Thousand Oaks, CA: Sage.
- LITTELL, R. C., MILLIKEN, G. A., STROUP, W. W., & WOLFINGER, R. R. (1996). *SAS system for mixed models*. Cary, NC: SAS Institute.
- NORMAND, S. (1995). Meta-analysis software: A comparative review. *American Statistician*, **49**, 298-309.
- PINHEIRO, J. C., & BATES, D. M. (2000). *Mixed-effects models in S and S-Plus*. New York: Springer-Verlag.
- RASBASH, J., BROWNE, W., GOLDSTEIN, H., YANG, M., ET AL. (2000). *A user's guide to MLwiN* (2nd ed.) [Computer software]. London: Institute of Education.
- RAUDENBUSH, S. W., BRYK, A. S., & CONGDON, R. T. (2000). *HLM 5: Hierarchical linear and nonlinear modeling* [Computer software]. Lincolnwood, IL: Scientific Software International.
- SAS INSTITUTE. (1999). *SAS/STAT user's guide, version 8* (Vol. 2). Cary, NC: Author.
- VERBEKE, G., & MOLENBERGHS, G. (Eds.). (1997). *Linear mixed models in practice: A SAS-oriented approach*. New York: Springer-Verlag.
- WANG, M. C., & BUSHMAN, B. J. (1999). *Integrating results through meta-analytic review using SAS software*. Cary, NC: SAS Institute.
- WARE, J. H. (1985). Linear models for the analysis of longitudinal studies. *American Statistician*, **39**, 95-101.

#### NOTES

1. However, in reality, this assumption is unlikely to be met, owing to the censoring effects of null findings. The issue of publication bias is complicated and beyond the scope of this paper. Refer to Begg (1994) for a more complete discussion on this topic.
2. It should be pointed out that simple noniterative formulas are available that can easily be used to perform a meta-analysis, using a simple spreadsheet (such as Excel) or a hand calculator (see, e.g., DerSimonian & Laird, 1986; Hedges & Olkin, 1985).
3. Using the `gdata` statement is another option, as was suggested by Wang and Bushman (1999).

(Manuscript received November 16, 2000;  
revision accepted for publication March 12, 2001.)